1 Lagrange equations - Example 1

The planar mechanical system considered is shown in Fig. 1.1. has a slider 1 of mass m and a pendulum 2 with the mass M concentrated at B. The length of AB is L and the elastic constant of the spring R is k. The spring deflect only horizontally.

Given the initial conditions find the equations of motion using Lagrange method.

The generalized coordinates for this two degree of freedom system are the displacement of the slider $q_1(t)$ and the rotation of the pendulum $q_2(t)$

Position analysis

A cartesian reference frame $xOyz$ with the versors [$\mathbf{\imath}$, $\mathbf{\jmath}$, $\mathbf{k}$] is selected, Fig. 1.1. The position vector of mass 1 is

$$\mathbf{r}_1 = \mathbf{r}_A = q_1(t)\,\mathbf{\imath}. \tag{1}$$

The position vector of mass 2 is

$$\mathbf{r}_2 = \mathbf{r}_B = [q_1(t) + L\sin q_2(t)]\,\mathbf{\imath} + L\cos q_2(t)\mathbf{\jmath}. \tag{2}$$

Velocity analysis

The velocity of the slider 1 is

$$\mathbf{v}_A = \frac{d\,\mathbf{r}_A}{d\,t} = \dot{\mathbf{r}}_A = \dot{q}_1\,\mathbf{\imath}, \tag{3}$$

and the velocity of the particle at B is

$$\mathbf{v}_B = \frac{d\,\mathbf{r}_B}{d\,t} = \dot{\mathbf{r}}_B = (\dot{q}_1 + L\dot{q}_2\cos q_2)\,\mathbf{\imath} - L\dot{q}_2\sin q_2\mathbf{\jmath}. \tag{4}$$

Kinetic energy

The kinetic energy of the slider 1 is

$$T_1 = \frac{1}{2}m\mathbf{v}_A\cdot\mathbf{v}_A = \frac{1}{2}m\dot{q}_1{}^2, \tag{5}$$

and the the kinetic energy of the mass 2 is

$$T_2 = \frac{1}{2}M\mathbf{v}_B\cdot\mathbf{v}_B = \frac{1}{2}M\left(\dot{q}_1{}^2 + 2L\dot{q}_1\dot{q}_2\cos q_2 + L^2\dot{q}_2{}^2\right). \tag{6}$$

The total kinetic energy is

$$T = T_1 + T_2. \tag{7}$$

External forces analysis

The forces that act on 1 at A are the spring force and the gravity force

$$\mathbf{F}_A = -kq_1\,\boldsymbol{\imath} + mg\,\boldsymbol{\jmath}, \tag{8}$$

where g=9.81 m/s^2 is the gravity acceleration. The gravity force acts on mass 2 at B

$$\mathbf{F}_B = Mg\,\boldsymbol{\jmath}. \tag{9}$$

Generalized forces

There are two generalized forces. The generalized force associated to q_1 is

$$\begin{aligned} &Q_1 = \mathbf{F}_A \cdot \frac{\partial \mathbf{r}_A}{\partial q_1} + \mathbf{F}_B \cdot \frac{\partial \mathbf{r}_B}{\partial q_1} = \\ &(-kq_1\,\boldsymbol{\imath} + mg\,\boldsymbol{\jmath}) \cdot \boldsymbol{\imath} + Mg\,\boldsymbol{\jmath} \cdot \boldsymbol{\imath} = -kq_1. \end{aligned} \tag{10}$$

The generalized force associated to q_2 is

$$\begin{aligned} &Q_2 = \mathbf{F}_A \cdot \frac{\partial \mathbf{r}_A}{\partial q_2} + \mathbf{F}_B \cdot \frac{\partial \mathbf{r}_B}{\partial q_2} = \\ &(-kq_1\,\boldsymbol{\imath} + mg\,\boldsymbol{\jmath}) \cdot \mathbf{0} + Mg\,\boldsymbol{\jmath} \cdot (L\cos q_2\,\boldsymbol{\imath} - L\sin q_2\,\boldsymbol{\jmath}) = \\ &-MgL\sin q_2. \end{aligned} \tag{11}$$

Lagrange equations

The two Lagrange equations are

$$\begin{aligned} \frac{d}{dt}\left(\frac{\partial T}{\partial \dot{q}_1}\right) - \frac{\partial T}{\partial q_1} &= Q_1, \\ \frac{d}{dt}\left(\frac{\partial T}{\partial \dot{q}_2}\right) - \frac{\partial T}{\partial q_2} &= Q_2. \end{aligned} \tag{12}$$

One can calculate for q_1

$$\begin{aligned} &\frac{\partial T}{\partial \dot{q}_1} = (m+M)\dot{q}_1 + LM\dot{q}_2\cos q_2, \\ &\frac{d}{dt}\left(\frac{\partial T}{\partial \dot{q}_1}\right) = (m+M)\ddot{q}_1 + LM\ddot{q}_2\cos q_2 - LM\dot{q}_2^{\,2}\sin q_2, \\ &\frac{\partial T}{\partial q_1} = 0. \end{aligned} \tag{13}$$

For the generalized coordinate q_2 the left hand side of Lagrange equation is

$$\begin{aligned}
&\frac{\partial T}{\partial \dot{q}_2} = LM\left(\dot{q}_1 \cos q_2 + L\dot{q}_2\right),\\
&\frac{d}{d\,t}\left(\frac{\partial T}{\partial \dot{q}_2}\right) = LM\left(\ddot{q}_1 \cos q_2 - \dot{q}_1 \dot{q}_2 \sin q_2 + L\ddot{q}_2\right),\\
&\frac{\partial T}{\partial q_2} = -LM\dot{q}_1\dot{q}_2 \sin q_2.
\end{aligned} \tag{14}$$

The equations of motion are

$$\begin{aligned}
&(m+M)\ddot{q}_1 + LM\ddot{q}_2 \cos q_2 - LM{\dot{q}_2}^2 \sin q_2 = -kq_1,\\
&LM\left(\ddot{q}_1 \cos q_2 - \dot{q}_1\dot{q}_2 \sin q_2 + L\ddot{q}_2\right) + LM\dot{q}_1\dot{q}_2 \sin q_2 = -MgL \sin q_2,
\end{aligned} \tag{15}$$

or

$$\begin{aligned}
&(m+M)\ddot{q}_1 + LM\ddot{q}_2 \cos q_2 - LM{\dot{q}_2}^2 \sin q_2 + kq_1 = 0,\\
&LM\ddot{q}_1 \cos q_2 + ML^2\ddot{q}_2 + MgL \sin q_2 = 0.
\end{aligned} \tag{16}$$

For small oscillations of the pendulum $\sin q_2 \approx q_2$ and $\cos q_2 \approx 1$, the equations of motion are

$$\begin{aligned}
&(m+M)\ddot{q}_1 + LM\ddot{q}_2 - LM{\dot{q}_2}^2 q_2 + kq_1 = 0,\\
&LM\ddot{q}_1 + ML^2\ddot{q}_2 + MgLq_2 = 0.
\end{aligned} \tag{17}$$



Figure 1

```
(*  Lagrange equations - Example 1   *)

Off[General::spell1];
Off[General::spell];
rA={q1[t],0,0};
rB={q1[t]+L*Sin[q2[t]],
    L*Cos[q2[t]],0};
vA=D[rA,t];
vB=D[rB,t];
T1=m vA.vA/2;
T2=M vB.vB/2;
T=T1+T2;
(*External forces*)
FA={-k q1[t],m g,0};
FB={0,M g,0};
(*Generalized forces*)
Q1=FA.D[rA,q1[t]]+FB.D[rB,q1[t]];
Q2=FA.D[rA,q2[t]]+FB.D[rB,q2[t]];
(*Lagrange's Equations*)
eq1=D[D[T,q1'[t],t]]-D[T,q1[t]]-Q1;
eq2=D[D[T,q2'[t],t]]-D[T,q2[t]]-Q2;
(*Small oscilations*)
rule={Sin[q2[t]]->q2[t],
      Cos[q2[t]]->1};
(*input data*)
rule1={m->1.,M->1.,L->1.,
       k->1.,g->10.};
equation1=eq1/.rule/.rule1;
equation2=eq2/.rule/.rule1;
sol=NDSolve[
  {equation1==0,equation2==0,
  q1[0]==.1,q2[0]==.1,
  q1'[0]==0.,q2'[0]==0.},
  {q1,q2},{t,0.,1.}];
Plot[Evaluate[q1[t]]/.sol,
 {t,0.,1.}, PlotRange->All,
     AxesLabel->{"t[s]","q1[m]"}];
Plot[Evaluate[q2[t]]/.sol,
{t,0.,1.}, PlotRange->All,
     AxesLabel->{"t[s]","q2[o]"}]
```